Lesson 18: Elementary antiderivatives for logarithmic functions

```
> restart;
```

Integrating a logarithmic function

The procedure for integrating rational functions is guaranteed to always work: it always produces an elementary antiderivative for the rational function.

Now I want to look at one case where there may or may not be an elementary antiderivative: a function that is rational except for one logarithmic term. The logarithmic term will be $\theta = \ln(R(x))$ where $R(x)$ is a rational function. I'll suppose our integrand is $f(x) = \dfrac{A(\theta)}{B(\theta)}$ where $A(\theta)$ and $B(\theta)$ are polynomials with coefficients that can be rational functions of x, with $A(\theta)$ of lower degree than $B(\theta)$, and $B(\theta)$ is square-free, i.e. it can't be divided by the square of any non-constant polynomial. Here are some examples:

- $\dfrac{1}{x\ln(x)}$ with $\theta = \ln(x)$, $A(\theta) = 1$, $B(\theta) = x\,\theta$. This does have an elementary antiderivative, and a Math 101 student might be able to find it.

```
> int(1/(x*ln(x)), x);
```

$$\ln(\ln(x)) \qquad \textbf{(1.1)}$$

- $\dfrac{x}{\ln(x)}$ with $\theta = \ln(x)$, $A(\theta) = x$, $B(\theta) = \theta$. This does not have an elementary antiderivative.

```
> int(x/ln(x), x);
```

$$-\mathrm{Ei}(1, -2\ln(x)) \qquad \textbf{(1.2)}$$

- $\dfrac{(x^2+1)\ln(x^2+1) - 2x^2}{(x^2+1)\left(\ln(x^2+1)^2 - x^2\right)}$ with $\theta = \ln(x^2+1)$, $A(\theta) = (x^2+1)\,\theta - 2x^2$, $B(\theta) = (x^2+1)\,(\theta^2 - x^2)$. This one is not so obvious, but

```
> int(((x^2+1)*ln(x^2+1) - 2*x^2)/((x^2+1)*(ln(x^2+1)^2 - x^2)
  ), x);
```

$$-\frac{1}{2}\ln\left(x - \ln(x^2+1)\right) + \frac{1}{2}\ln\left(x + \ln(x^2+1)\right) \qquad \textbf{(1.3)}$$

The method is somewhat analogous to what we did for the "logarithmic part" of rational functions: $\dfrac{p(x)}{q(x)}$ where $q(x)$ was square-free and had higher degree than $p(x)$. There we looked at the resultant of $q(x)$ and $p(x) - z\,q'(x)$ with respect to x, which is a polynomial in z with constant coefficients, and we found an antiderivative that was the sum of $r_j \ln\left(G_j(x)\right)$ for all roots r_j of that resultant, where $G_j(x) = gcd\left(q(x), p(x) - r_j\,q'(x)\right)$.

Here, the Rothstein-Trager Theorem says

- Let $R(z)$ be the resultant of $B(\theta)$ and $A(\theta) - z\,\frac{d}{dx}B(\theta)$ with respect to θ. Note that this is a polynomial in z, whose coefficients can be rational functions in x.
- $\frac{A(\theta)}{B(\theta)}$ has an elementary antiderivative if and only if all the roots of $R(z)$ are constants (i.e. don't depend on x).
- If they are constants, let $r_1, \ldots, r_n$ be those roots, and $G_j(x) = gcd\left(B(\theta), A(\theta) - r_j \frac{d}{dx}B(\theta)\right)$ (as polynomials in θ). Then
  $\int \frac{A(\theta)}{B(\theta)}\,dx = \sum_{j=1}^{n} r_j \ln(G_j(x))$.

One more point. In the **gcd** command, we don't say what the variable is, even if there are two possibilities (x and θ): **gcd** actually works for polynomials in several variables. But sometimes we might have polynomials in θ with coefficients that are rational functions (but not polynomials) in x. Then **gcd** would complain:

```
> gcd(theta/x, theta^2 + theta);
Error, (in gcd/Freeze) arguments should be polynomials
```

The cure for this is to use **gcdex** instead. The **ex** stands for extended. You give **gcdex** the variable name as well as the two polynomials:

```
> gcdex(theta/x, theta^2 + theta, theta);
```

$$\theta \tag{1.4}$$

Let's see how this works in each of the examples I mentioned.

- Example 1: $\frac{1}{x\ln(x)}$ with $\theta = \ln(x)$, $A(\theta) = 1$, $B(\theta) = x\,\theta$

```
> A:= 1; B:= x*ln(x);
  C:= A - z*diff(B, x);
```

$$A := 1$$

$$B := x\ln(x)$$

$$C := 1 - z\,(\ln(x) + 1) \tag{1.5}$$

```
> R:=resultant(x*theta, 1-z*(theta+1), theta);
```

$$R := x\,(1 - z) \tag{1.6}$$

```
> solve(R,z);
```

$$1 \tag{1.7}$$

That's constant, so yes, this has an elementary antiderivative. There's only one G.

```
> G:= gcdex(x*theta, 1-1*(theta+1),theta);
```

$$G := \theta \tag{1.8}$$

And the antiderivative is:

```
> F := eval(1*ln(G),theta=ln(x));
```

$$F := \ln(\ln(x)) \tag{1.9}$$

• Example 2: $\dfrac{x}{\ln(x)}$ with $\theta = \ln(x)$, $A(\theta) = x$, $B(\theta) = \theta$.

```
> A:= x; B:= ln(x);
  C:= A - z*diff(B, x);
```

$$A := x$$

$$B := \ln(x)$$

$$C := x - \frac{z}{x} \tag{1.10}$$

```
> R:=resultant(theta, x - z/x, theta);
```

$$R := \frac{x^2 - z}{x} \tag{1.11}$$

```
> solve(R,z);
```

$$x^2 \tag{1.12}$$

That's not constant, so the answer is no: this one doesn't have an elementary antiderivative.

• Example 3: $\dfrac{(x^2+1)\ln(x^2+1) - 2x^2}{(x^2+1)\left(\ln(x^2+1)^2 - x^2\right)}$ with $\theta = \ln(x^2+1)$, $A(\theta) = (x^2+1)\,\theta - 2x^2$, $B(\theta) = (x^2+1)\,(\theta^2 - x^2)$.

```
> A:= (x^2+1)*ln(x^2+1)-2*x^2;
  B:= (x^2+1)*(ln(x^2+1)^2-x^2);
  C:= A - z*diff(B,x);
```

$$A := (x^2+1)\ln(x^2+1) - 2x^2$$

$$B := (x^2+1)\left(\ln(x^2+1)^2 - x^2\right)$$

$$C := (x^2+1)\ln(x^2+1) - 2x^2 - z\left(2x\left(\ln(x^2+1)^2 - x^2\right) + (x^2+1)\left(\frac{4\ln(x^2+1)\,x}{x^2+1} - 2x\right)\right) \tag{1.13}$$

```
> Atheta:= eval(A,ln(x^2+1)=theta);
  Btheta:= eval(B,ln(x^2+1)=theta);
  Ctheta:= eval(C,ln(x^2+1)=theta);
```

$$Atheta := (x^2+1)\,\theta - 2x^2$$

$$Btheta := (x^2+1)\,(\theta^2 - x^2)$$

$$Ctheta := (x^2+1)\,\theta - 2x^2 - z\left(2x\,(\theta^2 - x^2) + (x^2+1)\left(\frac{4\,\theta\,x}{x^2+1} - 2x\right)\right) \tag{1.14}$$

```
> R:=resultant(Btheta, Ctheta, theta);
```

$$R := (x^2+1)^2\,x^2\,(-1 + 4x^4z^2 - 8x^2z^2 + 4z^2 + 2x^2 - x^4) \tag{1.15}$$

```
> solve(R,z);
```

$$\frac{1}{2}, -\frac{1}{2} \tag{1.16}$$

Both are constants, so the answer is yes.

```
> G1 := gcdex(Btheta, eval(Ctheta, z=1/2),theta);
```

$$G1 := \theta + x \tag{1.17}$$

```
> G2 := gcdex(Btheta, eval(Ctheta, z=-1/2),theta);
```

$$G2 := \theta - x \tag{1.18}$$

And the antiderivative is:

```
> F:= eval(1/2*ln(G1)-1/2*ln(G2), theta=ln(x^2+1));
```

$$F := \frac{1}{2}\ln\left(x + \ln\left(x^2+1\right)\right) - \frac{1}{2}\ln\left(\ln\left(x^2+1\right) - x\right) \tag{1.19}$$

```
> normal(diff(F,x) - A/B);
```

$$0 \tag{1.20}$$

infolevel

When you want some idea of how Maple does some computation, the **infolevel** facility may help. For information on a particular command **f**, you give a value (an integer from 1 to 5) to **infolevel[f]**. For example, for information on integration you could try

```
> infolevel[int] := 3;
```

$$infolevel_{int} := 3$$

```
> int(1/(1+x^2+x^5),x);
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
int/ratpoly/horowitz: integrating
```

$$\frac{1}{1 + \_X^2 + \_X^5}$$

```
int/ratpoly/horowitz: Horowitz' method yields
```

$$\int \frac{1}{1 + \_X^2 + \_X^5}\, \mathrm{d}\_X$$

```
int/risch/ratpoly: starting computing subresultants at time
1.014
int/risch/ratpoly: end of subresultants computation at time
1.029
int/risch/ratpoly: Rothstein's method - factored resultant is
```

$$\left[\left[z^5 + \frac{27}{3233}\, z^3 + \frac{50}{3233}\, z^2 - \frac{1}{3233},\, 1\right]\right]$$

```
int/risch/ratpoly: result is
```

$$\sum_{\_R = RootOf\left(3233\,\_Z^5 + 27\,\_Z^3 + 50\,\_Z^2 - 1\right)} \_R \ln\left(\_X - \frac{2224698426}{7363759}\,\_R^4 + \frac{797015325}{7363759}\,\_R^3 - \frac{109507419}{7363759}\,\_R^2 + \frac{18837605}{7363759}\,\_R + \frac{7092000}{7363759}\right)$$

$$\sum_{\_R=RootOf(3233\_Z^5+27\_Z^3+50\_Z^2-1)} \_R \ln\left(x - \frac{2224698426}{7363759}\_R^4 + \frac{797015325}{7363759}\_R^3 - \frac{109507419}{7363759}\_R^2 + \frac{18837605}{7363759}\_R + \frac{7092000}{7363759}\right)$$

The basic scheme is:

Level 1: reserved for information that the user must be told.

Level 2,3: general information, including technique or algorithm being used.

Level 4,5: more detailed information about how the problem is being solved

However, the usefulness of this facility is rather uneven: it's up to the Maple programmers to decide what to tell, and this varies a lot from one command to another.

By the way, in many cases (such as **int**), if you try setting the infolevel and then repeating a command you already tried, you won't get anything from infolevel, because Maple just remembers the result it got last time rather than doing it all over again. In such a case you might want to use **restart** to get rid of all remembered results.

```
> int(1/(1+x^2+x^5),x);
```

$$\sum_{\_R=RootOf(3233\_Z^5+27\_Z^3+50\_Z^2-1)} \_R \ln\left(x - \frac{2224698426}{7363759}\_R^4 + \frac{797015325}{7363759}\_R^3 - \frac{109507419}{7363759}\_R^2 + \frac{18837605}{7363759}\_R + \frac{7092000}{7363759}\right) \quad \textbf{(2.1)}$$

```
> restart; infolevel[int]:= 1; int(1/(1+x^2+x^5),x);
```

$$infolevel_{int} := 1$$

```
int/indef1: first-stage indefinite integration
int/ratpoly: rational function integration
```

$$\sum_{\_R=RootOf(3233\_Z^5+27\_Z^3+50\_Z^2-1)} \_R \ln\left(x - \frac{2224698426}{7363759}\_R^4 + \frac{797015325}{7363759}\_R^3 - \frac{109507419}{7363759}\_R^2 + \frac{18837605}{7363759}\_R + \frac{7092000}{7363759}\right)$$

If you don't know which command you need to know about, you might try **infolevel[all]**, which will apply to all Maple commands.

```
> infolevel[all]:= 1;
  int(x*ln(x)/(1+ln(x)^2), x);
```

$$infolevel_{all} := 1$$

```
int/indef1: first-stage indefinite integration
int/indef2: second-stage indefinite integration
int/ln: case of integrand containing ln
int/rischnorm: enter Risch-Norman integrator
```

```
radfield: computing a basis for  {}
radfield: over  {}
radfield: options are {}
radfield: computing a basis for  {}
radfield: over  {}
radfield: options are {}
factor/polynom: polynomial factorization: number of terms 2
factor/polynom: polynomial factorization: number of terms 2
solve: Warning: no solutions found
int/rischnorm: exit Risch-Norman integrator
int/risch: enter Risch integration
radfield: computing a basis for  {}
radfield: over  {}
radfield: options are {}
radfield: computing a basis for  {}
radfield: over  {}
radfield: options are {}
radfield: computing a basis for  {}
radfield: over  {}
radfield: options are {}
radfield: computing a basis for  {}
radfield: over  {}
radfield: options are {}
int/risch: exit Risch integration
```

$$\int \frac{x \ln(x)}{1 + \ln(x)^2} \, dx$$

To turn off the use of infolevel for a particular command, you can set it to 0.

```
> infolevel[all] := 0; infolevel[int] := 0;
```

$$infolevel_{all} := 0$$

$$infolevel_{int} := 0$$

```
> int(1/(x^11+1),x);
```

$$\frac{1}{11} \sum_{\_R = RootOf(\_Z^{10} + \_Z^9 + \_Z^8 + \_Z^7 + \_Z^6 + \_Z^5 + \_Z^4 + \_Z^3 + \_Z^2 + \_Z + 1)} \_R \ln(x + \_R) + \frac{1}{11} \ln(x + 1)$$

Numerical integration

Maple has some very good numerical methods for calculating definite integrals. As we've seen, these will work even when **int** does not return a value.

```
> int(x/(sin(x)*(1+cos(x)^2)),x);
```

$$\frac{1}{2}\,\mathrm{I}\,\mathrm{dilog}\left(1+\mathrm{e}^{\mathrm{I}x}\right)-\frac{\frac{1}{4}\,\mathrm{I}\sqrt{2}\,x\ln\left(-\frac{\mathrm{I}\left(\mathrm{I}\left(\sqrt{2}-1\right)-\mathrm{e}^{\mathrm{I}x}\right)}{\sqrt{2}-1}\right)}{\sqrt{2}-1} \tag{3.1}$$

$$+\frac{1}{4}\,\frac{\sqrt{2}\,\mathrm{dilog}\left(-\frac{\mathrm{I}\left(\mathrm{I}\left(1+\sqrt{2}\right)-\mathrm{e}^{\mathrm{I}x}\right)}{1+\sqrt{2}}\right)}{1+\sqrt{2}}+\frac{\frac{1}{4}\,\mathrm{I}\,x\ln\left(-\frac{\mathrm{I}\left(\mathrm{I}\left(1+\sqrt{2}\right)-\mathrm{e}^{\mathrm{I}x}\right)}{1+\sqrt{2}}\right)}{1+\sqrt{2}}$$

$$-\frac{1}{4}\,\frac{\mathrm{dilog}\left(\frac{\mathrm{I}\left(-\mathrm{I}\left(1+\sqrt{2}\right)-\mathrm{e}^{\mathrm{I}x}\right)}{1+\sqrt{2}}\right)}{1+\sqrt{2}}+\frac{\frac{1}{4}\,\mathrm{I}\sqrt{2}\,x\ln\left(-\frac{\mathrm{I}\left(\mathrm{I}\left(1+\sqrt{2}\right)-\mathrm{e}^{\mathrm{I}x}\right)}{1+\sqrt{2}}\right)}{1+\sqrt{2}}$$

$$-\frac{1}{4}\,\frac{\sqrt{2}\,\mathrm{dilog}\left(\frac{\mathrm{I}\left(-\mathrm{I}\left(1+\sqrt{2}\right)-\mathrm{e}^{\mathrm{I}x}\right)}{1+\sqrt{2}}\right)}{1+\sqrt{2}}-\frac{\frac{1}{4}\,\mathrm{I}\,x\ln\left(\frac{\mathrm{I}\left(-\mathrm{I}\left(1+\sqrt{2}\right)-\mathrm{e}^{\mathrm{I}x}\right)}{1+\sqrt{2}}\right)}{1+\sqrt{2}}$$

$$-\frac{1}{4}\,\frac{\sqrt{2}\,\mathrm{dilog}\left(-\frac{\mathrm{I}\left(\mathrm{I}\left(\sqrt{2}-1\right)-\mathrm{e}^{\mathrm{I}x}\right)}{\sqrt{2}-1}\right)}{\sqrt{2}-1}+\frac{\frac{1}{4}\,\mathrm{I}\,x\ln\left(-\frac{\mathrm{I}\left(\mathrm{I}\left(\sqrt{2}-1\right)-\mathrm{e}^{\mathrm{I}x}\right)}{\sqrt{2}-1}\right)}{\sqrt{2}-1}$$

$$-\frac{1}{4}\,\frac{\mathrm{dilog}\left(\frac{\mathrm{I}\left(-\mathrm{I}\left(\sqrt{2}-1\right)-\mathrm{e}^{\mathrm{I}x}\right)}{\sqrt{2}-1}\right)}{\sqrt{2}-1}-\frac{\frac{1}{4}\,\mathrm{I}\sqrt{2}\,x\ln\left(\frac{\mathrm{I}\left(-\mathrm{I}\left(1+\sqrt{2}\right)-\mathrm{e}^{\mathrm{I}x}\right)}{1+\sqrt{2}}\right)}{1+\sqrt{2}}$$

$$+\frac{1}{4}\,\frac{\sqrt{2}\,\mathrm{dilog}\left(\frac{\mathrm{I}\left(-\mathrm{I}\left(\sqrt{2}-1\right)-\mathrm{e}^{\mathrm{I}x}\right)}{\sqrt{2}-1}\right)}{\sqrt{2}-1}+\frac{1}{2}\,\mathrm{I}\,\mathrm{dilog}\left(\mathrm{e}^{\mathrm{I}x}\right)$$

$$+\frac{1}{4}\,\frac{\mathrm{dilog}\left(-\frac{\mathrm{I}\left(\mathrm{I}\left(1+\sqrt{2}\right)-\mathrm{e}^{\mathrm{I}x}\right)}{1+\sqrt{2}}\right)}{1+\sqrt{2}}-\frac{\frac{1}{4}\,\mathrm{I}\,x\ln\left(\frac{\mathrm{I}\left(-\mathrm{I}\left(\sqrt{2}-1\right)-\mathrm{e}^{\mathrm{I}x}\right)}{\sqrt{2}-1}\right)}{\sqrt{2}-1}$$

$$-\frac{1}{2}\,x\ln\left(1+\mathrm{e}^{\mathrm{I}x}\right)+\frac{\frac{1}{4}\,\mathrm{I}\sqrt{2}\,x\ln\left(\frac{\mathrm{I}\left(-\mathrm{I}\left(\sqrt{2}-1\right)-\mathrm{e}^{\mathrm{I}x}\right)}{\sqrt{2}-1}\right)}{\sqrt{2}-1}$$

$$+\frac{1}{4}\,\frac{\mathrm{dilog}\left(-\frac{\mathrm{I}\left(\mathrm{I}\left(\sqrt{2}-1\right)-\mathrm{e}^{\mathrm{I}x}\right)}{\sqrt{2}-1}\right)}{\sqrt{2}-1}$$

```
> infolevel[evalf]:= 3;
  evalf(Int(x/(sin(x)*(1+cos(x)^2)), x= 0..Pi/2));
```

$$infolevel_{evalf} := 3$$

```
evalf/int/control: integrating on 0 .. 1/2*Pi the integrand
```

$$\frac{x}{\sin(x)\left(1+\cos(x)^2\right)}$$

```
evalf/int/control: tolerance = .5000000000e-9; method =
_DEFAULT; maxintervals = _DEFAULT
Control: Entering NAGInt
Control: trying d01ajc (nag_1d_quad_gen)
d01ajc: epsabs=.500000000000000e-12; epsrel=.5000000000e-9;
max_num_subint=200
d01ajc: result=1.33861101927154235
d01ajc: abserr=.628550758933998994e-11; num_subint=1;
fun_count=21
Control: result=1.33861101927154235
```

$$1.338611019$$

NAG stands for Numerical Algorithms Group, which wrote lots of the numerical algorithms that Maple uses, especially those that use "hardware" floating point, i.e. the built-in binary floating-point arithmetic that comes with the computer, rather than Maple's usual arbitrary-precision decimal "software" floating point. Hardware floating-point arithmetic is much faster, and in particular the NAG algorithms are highly optimized and very fast. In this case Maple used a NAG algorithm called **d01ajc**. It returned a hardware-float result of 1.33861101927154235, which Maple then rounded to 10 digits as
1.338611019.

```
> evalhf(Digits);
```

$$15.$$

That's roughly the Digits equivalent of hardware floats on this machine. If we asked for more than that, Maple would not use the NAG algorithm.

```
> evalf(Int(x/(sin(x)*(1+cos(x)^2)), x= 0..Pi/2),16);
evalf/int/control: integrating on 0 .. 1/2*Pi the integrand
```

$$\frac{x}{\sin(x)\,\left(1+\cos(x)^2\right)}$$

```
evalf/int/control: tolerance = .5000000000000000e-15; method =
_DEFAULT; maxintervals = _DEFAULT
evalf/int/CreateProc: Trying makeproc
evalf/int/ccquad: n = 2 integral estimate =
1.317562566127062387
                    n = 6 integral estimate =
1.338600467181040950
evalf/int/ccquad: n = 18 integral estimate =
1.338611019271054250
                                  error =
.1562023575177172825e-10
evalf/int/ccquad: n = 54 integral estimate =
1.338611019271379933
                                  error =
.1338611019271379933e-16
From ccquad, result = 1.338611019271379933 integrand evals =
```

```
55 error = .1338611019271379933e-16
tolerance = .6693055096356899665e-15
```

$$1.338611019271380$$

This time it used a non-NAG method called **ccquad** (Clenshaw-Curtis quadrature). You can get some information on these methods using Maple's help:

```
> ?evalf,int
```

For the NAG methods, there's lots of information at the NAG web site .

Maple uses some quite sophisticated methods for numerical approximation of integrals. We'll look at some rather less sophisticated ones, to get an idea of what is behind some of these.

We'll start with some that you may remember from Math 101: left and right Riemann sums, the Midpoint Rule, Trapezoid Rule and Simpson's Rule.

Midpoint and Trapezoid

```
> restart;
```

Suppose we want to approximate $J = \int_a^b f(x)\, dx$ numerically. The simplest way is called a **Riemann sum**: you divide the interval $[a, b]$ into n equal subintervals and approximate $f(x)$ on each subinterval by its value at one point of the subinterval.

$J \approx \sum_{j=1}^{n} f(x_j)\, h$ where $h = \frac{(b-a)}{n}$.

We'll usually have a, b and f fixed, and look at different n's, so I'll write everything as a function of n.

The step size h is the length of each subinterval.

```
> h := n -> (b-a)/n;
```

$$h := n \to \frac{b-a}{n}$$

The k'th interval goes from x_{k-1} to x_k, where $x_k = a + k\, h$.

```
> X := (k,n) -> a + k*h(n);
```

$$X := (k, n) \to a + k\, h(n)$$

Calculus text authors seem to like "left sums", where you take the value at the left end of the subinterval, and "right sums", where you take the value at the right end.

```
> LeftSum:= n -> add(f(X(k-1,n))*h(n), k=1..n);
```

$$LeftSum := n \to add(f(X(k-1, n))\, h(n), k = 1 ..n) \qquad \textbf{(4.1)}$$

```
> eval(LeftSum(4), {a=0,b=1});
```

$$\frac{1}{4} f(0) + \frac{1}{4} f\left(\frac{1}{4}\right) + \frac{1}{4} f\left(\frac{1}{2}\right) + \frac{1}{4} f\left(\frac{3}{4}\right) \qquad \textbf{(4.2)}$$

```
> RightSum:= n -> add(f(X(k,n))*h(n), k=1..n);
```

$$RightSum := n \to add(f(X(k, n))\, h(n), k = 1 ..n) \qquad \textbf{(4.3)}$$

```
> eval(RightSum(4), {a=0,b=1});
```

$$\frac{1}{4} f\left(\frac{1}{4}\right) + \frac{1}{4} f\left(\frac{1}{2}\right) + \frac{1}{4} f\left(\frac{3}{4}\right) + \frac{1}{4} f(1) \quad \textbf{(4.4)}$$

A better idea would be to go halfway along the subinterval. The Midpoint Rule approximation to the integral is $M_n = \sum_{k=1}^{n} f\left(\frac{x_{k-1} + x_k}{2}\right) h$.

```
> M := n -> add(f((X(k-1,n)+X(k,n))/2)*h(n), k=1..n);
```

$$M := n \rightarrow add\left(f\left(\frac{1}{2} X(k-1, n) + \frac{1}{2} X(k, n)\right) h(n), k = 1..n\right)$$

```
> eval(M(4), {a=0,b=1});
```

$$\frac{1}{4} f\left(\frac{1}{8}\right) + \frac{1}{4} f\left(\frac{3}{8}\right) + \frac{1}{4} f\left(\frac{5}{8}\right) + \frac{1}{4} f\left(\frac{7}{8}\right) \quad \textbf{(4.5)}$$

```
> M(n);
Error, (in M) unable to execute add
```

The **add** function needs the range to be specified explicitly, not using a symbolic variable n. Here's a "formal" version of **M**.

```
> Mformal := n -> Sum(f((X(k-1,n)+X(k,n))/2)*h(n), k=1..n);
  eval(Mformal(n),{a=0,b=1});
```

$$Mformal := n \rightarrow \sum_{k=1}^{n} f\left(\frac{1}{2} X(k-1, n) + \frac{1}{2} X(k, n)\right) h(n)$$

$$\sum_{k=1}^{n} \frac{f\left(\frac{1}{2} \frac{k-1}{n} + \frac{1}{2} \frac{k}{n}\right)}{n}$$

For the Trapezoid Rule, we approximate $f(x)$ on the subinterval by the average of its values at the two endpoints x_{k-1} and x_k.

```
> T := n -> add((f(X(k-1,n)) + f(X(k,n)))/2 * h(n), k=1..n);
  Tformal := n -> Sum((f(X(k-1,n)) + f(X(k,n)))/2 * h(n), k=1..
  n);
```

$$T := n \rightarrow add\left(\frac{1}{2} \left(f(X(k-1, n)) + f(X(k, n))\right) h(n), k = 1..n\right)$$

$$Tformal := n \rightarrow \sum_{k=1}^{n} \frac{1}{2} \left(f(X(k-1, n)) + f(X(k, n))\right) h(n)$$

```
> Tformal(n);
```

$$\sum_{k=1}^{n} \frac{1}{2} \frac{\left(f\left(a + \frac{(k-1)(b-a)}{n}\right) + f\left(a + \frac{k(b-a)}{n}\right)\right)(b-a)}{n}$$

```
> eval(T(4), {a=0, b=1});
```

$$\frac{1}{8} f(0) + \frac{1}{4} f\left(\frac{1}{4}\right) + \frac{1}{4} f\left(\frac{1}{2}\right) + \frac{1}{4} f\left(\frac{3}{4}\right) + \frac{1}{8} f(1)$$

Let's try a typical function f for which we know the true value of the integral, and look at the error (the difference between the true value and the approximation) for Midpoint and Trapezoid Rules with different values of n.

```
> a := 0: b := 1: f := x -> 1/(x + 1):
  J := int(f(x),x=a..b);
```

$$J := \ln(2) \quad (4.6)$$

```
> LeftSumErrors := [seq([n, evalf(J - LeftSum(n))], n = 1.. 20)
  ];
  RightSumErrors := [seq([n, evalf(J - RightSum(n))], n = 1..
  20)];
```

LeftSumErrors := [[1, -0.3068528194], [2, -0.1401861527], [3, -0.0901861527], [4, -0.0663766289], [5, -0.0524877400], [6, -0.0433968309], [7, -0.0369865745], [8, -0.0322246698], [9, -0.0285481992], [10, -0.0256242226], [11, -0.0232432702], [12, -0.0212669856], [13, -0.0196003189], [14, -0.0181758175], [15, -0.0169442904], [16, -0.0158690216], [17, -0.0149220519], [18, -0.0140817158], [19, -0.0133309650], [20, -0.0126562012]]

RightSumErrors := [[1, 0.1931471806], [2, 0.1098138473], [3, 0.0764805139], [4, 0.0586233711], [5, 0.0475122600], [6, 0.0399365024], [7, 0.0344419969], [8, 0.0302753302], [9, 0.0270073564], [10, 0.0243757774], [11, 0.0222112753], [12, 0.0203996811], [13, 0.0188612195], [14, 0.0175384682], [15, 0.0163890429], [16, 0.0153809784], [17, 0.0144897128], [18, 0.0136960620], [19, 0.0129848244], [20, 0.0123437988]] **(4.7)**

```
> MidpointErrors := [seq([n, evalf(J - M(n))], n = 1 .. 20)];
```

MidpointErrors := [[1, 0.0264805139], [2, 0.0074328949], [3, 0.0033924908], [4, 0.0019272894], [5, 0.0012392949], [6, 0.0008628597], [7, 0.0006349395], [8, 0.0004866266], [9, 0.0003847676], [10, 0.0003118202], [11, 0.0002577997], [12, 0.0002166855], [13, 0.0001846727], [14, 0.0001592614], [15, 0.0001387542], [16, 0.0001219663], [17, 0.0001080498], [18, 0.0000963856], [19, 0.0000865128], [20, 0.0000780824]] **(4.8)**

```
> TrapezoidErrors := [seq([n, evalf(J - T(n))], n = 1 .. 20)];
```

TrapezoidErrors := [[1, -0.0568528194], [2, -0.0151861527], [3, -0.0068528194], [4, -0.0038766289], [5, -0.0024877400], [6, -0.0017301643], [7, -0.0012722888], [8, -0.0009746698], [9, -0.0007704214], [10, -0.0006242226], [11, -0.0005159975], [12, -0.0004336523], [13, -0.0003695497], [14, -0.0003186747], [15, -0.0002776238], [16, -0.0002440216], [17, -0.0002161696], [18, -0.0001928269], [19, -0.0001730703], [20, -0.0001562012]] **(4.9)**

Maple objects introduced in this lesson

gcdex
infolevel